Math 1. 2-hours test May 13, 2017

JE/JKL 13.5.17

Problem 1

```
> restart:with(plots):
```

A function f of two real variables is for $(x, y) \neq (0, 0)$ given by

```
> f:=(x,y)->y/(x^2+y^2);
```

$$f := (x, y) \rightarrow \frac{y}{x^2 + y^2} \quad (1.1)$$

```
> f(x,y);
```

$$\frac{y}{x^2 + y^2} \quad (1.2)$$

Question 1

In the (x, y)-plane the three points A= (0, 1), B = (0, -1) and C= $(\frac{1}{2}, \frac{1}{2})$ are considered.

```
> 'f(0,1)'=f(0,1);
```

$$f(0, 1) = 1 \quad (1.1.1)$$

```
> 'f(0,-1)'=f(0,-1);
```

$$f(0, -1) = -1 \quad (1.1.2)$$

```
> 'f(1/2,1/2)'=f(1/2,1/2);
```

$$f\left(\frac{1}{2}, \frac{1}{2}\right) = 1 \quad (1.1.3)$$

Since f(A) = f(C) =1, A and C lie on the same level curve for f viz. on the level curve

$f(x, y) = 1$, $(x, y) \neq (0, 0) \Leftrightarrow \frac{y}{x^2 + y^2} = 1$, $(x, y) \neq (0, 0) \Leftrightarrow$

$x^2 + y^2 - y = 0$, $(x, y) \neq (0, 0) \Leftrightarrow x^2 + \left(y - \frac{1}{2}\right)^2 = \frac{1}{4}$, $(x, y) \neq (0, 0)$,

that is, the circle with centre at $(0, \frac{1}{2})$ and radius $\frac{1}{2}$ except the point (0,0).

Since f(B) = -1, B does not lie on this level curve for f, but on the level curve

$f(x, y) = -1$, $(x, y) \neq (0, 0) \Leftrightarrow x^2 + \left(y + \frac{1}{2}\right)^2 = \frac{1}{4}$, $(x, y) \neq (0, 0)$,

that is, the circle with centre at $(0, -\frac{1}{2})$ and radius $\frac{1}{2}$ except the point (0,0).

The gradient for f is given by

$$\nabla f(x, y) = (f'_x(x, y), f'_y(x, y)) = \left(\frac{-2\,xy}{(x^2 + y^2)^2}, \frac{x^2 - y^2}{(x^2 + y^2)^2}\right), \ (x, y) \neq (0, 0).$$

Question 2

If the first coordinate has to be zero, then either x of y must be zero. This implies that if the second coordinate also has to be zero, then both x and y must be zero. But ∇f is not defind for (0,0). Since $\nabla f(x, y) \neq (0, 0)$ for all $(x, y) \neq (0, 0)$, f has no stationary points.

We consider the bounded and closed set of points $M = \{ (x, y) | \; 1 \leq x^2 + y^2 \leq 4 \}$.
I.e. M is the closed circular disc between the two discs $x^2 + y^2 = 1$ (centre $(0, 0)$ and radius 1) and $x^2 + y^2 = 4$ (centre $(0, 0)$ and radius 2).

```
> C1:=implicitplot(x^2+y^2<4,x=-2..2,y=-2..2,scaling=constrained,
  linestyle=1):
> C2:=implicitplot(x^2+y^2>1,x=-2..2,y=-2..2,filled=true,
  coloring=[gray,white],scaling=constrained,linestyle=1):
> R1:=implicitplot(x^2+y^2=1,x=-1..1,y=-1..1,color=black,scaling=
  constrained,linestyle=1,thickness=2):
> R2:=implicitplot(x^2+y^2=4,x=-2..2,y=-2..2,color=black,scaling=
  constrained,linestyle=1,thickness=2):
> C3:=implicitplot(x^2+y^2>4,x=-2..2,y=-2..2,scaling=constrained,
  filled=true,coloring=[white,white],linestyle=2):
> display(C1,C3,C2,R1,R2,title="Cirkelringen M");
```



Note that it is not a requirement to the answer that you can draw this figure of M in Maple.

Question 3

Since M is bounded and closed and since f is continuous in M, then f has a global minimum and a global maximum in M. Since f has neither stationary points nor exception points in the interior of M, these values are attained at the boudary of M.

Boundary investigation:

(1):

$x^2+y^2=1 \Leftrightarrow x=\pm\sqrt{1-y^2}\ ,\ y\in[-1;1]$.

$g(y)=f\left(\pm\sqrt{1-y^2},y\right)=y,\ y\in[-1;1]$.

$g'(y)=1>0$ for all $y\in]-1;1[$. I.e. g is increasing.

$g(-1)=f(0,-1)=-1$ is smallest and $g(1)=f(0,1)=1$ is largest.

(2):

$x^2+y^2=4 \Leftrightarrow x=\pm\sqrt{4-y^2}\ ,\ y\in[-2;2]$.

$h(y)=f\left(\pm\sqrt{4-y^2},y\right)=\frac{1}{4}y, y\in[-2;2]$.

$h'(y)=\frac{1}{4}>0$ for all $y\in]-2;2[$. I.e. h is increasing.

$h(-2)=f(0,-2)=-\frac{1}{2}$ is smallest and $h(2)=f(0,2)=\frac{1}{2}$ is largest.

By numerical comparison of these investigation we find that the global minimum is -1 and is attained at point B and that the global maximum is 1, that is attained in point A. Note that since M is connected, the range is $f(M)=[-1;1]$.

Problem 2

```
> restart:
```

The approximating second-degree polynomial about the point (0, 0) of a real function $f(x,y)$ is given by

$$P_2(x,y)=f(0,0)+f'_x(0,0)\,x+f'_y(0,0)\,y+\frac{1}{2}\left(f''_{xx}(0,0)x^2+2f''_{xy}(0,0)xy+f''_{yy}(0,0)y^2\right)$$

$$=2+\frac{1}{2}x^2+y^2=2+\frac{1}{2}\left(x^2+2y^2\right).$$

while the approximating second-degree polynomial about the point $\left(\frac{\sqrt{6}}{3},0\right)$ is given by

$$Q_2(x,y)=f\left(\frac{\sqrt{6}}{3},0\right)+f'_x\left(\frac{\sqrt{6}}{3},0\right)\left(x-\frac{\sqrt{6}}{3}\right)+f'_y\left(\frac{\sqrt{6}}{3},0\right)y$$

$$+\frac{1}{2}\left(f''_{xx}\left(\frac{\sqrt{6}}{3},0\right)\left(x-\frac{\sqrt{6}}{3}\right)^2+2f''_{xy}\left(\frac{\sqrt{6}}{3},0\right)\left(x-\frac{\sqrt{6}}{3}\right)y+f''_{yy}\left(\frac{\sqrt{6}}{3},0\right)y^2\right)$$

$$=\frac{4}{3}\sqrt{e}-\sqrt{e}\left(x-\frac{\sqrt{6}}{3}\right)^2+\frac{1}{3}\sqrt{e}\,y^2=\frac{4}{3}\sqrt{e}+\frac{1}{2}\left(-2\sqrt{e}\left(x-\frac{\sqrt{6}}{3}\right)^2+\frac{2}{3}\sqrt{e}\,y^2\right).$$

Question 1

From the given expression for $P_2(x,y)$ we read that

$f(0,0)=2$, $f'_x(0,0)=0$, $f'_y(0,0)=0$, $f''_{xx}(0,0)=1$, $f''_{xy}(0,0)=0$ and $f''_{yy}(0,0)=2$.

Question 2

If f has a proper local maximum or a proper local minimum in a point, then the point must be a stationary point, since f has no exception points.
Since $\nabla f(0,0) = (f'_x(0,0), f'_y(0,0)) = (0,0)$, (0,0) is a stationary point for f.
The Hessian matrix for f in the point (0,0) is

$$\mathbf{H}(0,0) = \begin{bmatrix} f''_{xx}(0,0) & f''_{xy}(0,0) \\ f''_{xy}(0,0) & f''_{yy}(0,0) \end{bmatrix} = \begin{bmatrix} 1 & 0 \\ 0 & 2 \end{bmatrix},$$

that has the eigenvalues 1 and 2.
Since both eigenvalues for $\mathbf{H}(0,0)$ are positive, then $f(0,0) = 2$ is a proper local minimum.

Question 3

From the given expression for $Q_2(x,y)$ we read that

$$f\left(\frac{\sqrt{6}}{3}, 0\right) = \frac{4}{3}\sqrt{e},$$

$$f'_x\left(\frac{\sqrt{6}}{3}, 0\right) = 0, f'_y\left(\frac{\sqrt{6}}{3}, 0\right) = 0, f''_{xx}\left(\frac{\sqrt{6}}{3}, 0\right) = -2\sqrt{e},\ f''_{xy}\left(\frac{\sqrt{6}}{3}, 0\right) = 0 \text{ and}$$

$$f''_{yy}\left(\frac{\sqrt{6}}{3}, 0\right) = \frac{2}{3}\sqrt{e}.$$

Since $\nabla f\left(\frac{\sqrt{6}}{3}, 0\right) = \left(f'_x\left(\frac{\sqrt{6}}{3}, 0\right), f'_y\left(\frac{\sqrt{6}}{3}, 0\right)\right) = (0,0)$, $\left(\frac{\sqrt{6}}{3}, 0\right)$ is also a stationary point for f.

The Hessian matrix for f in the point $\left(\frac{\sqrt{6}}{3}, 0\right)$ is

$$\mathbf{H}\left(\frac{\sqrt{6}}{3}, 0\right) = \begin{bmatrix} f''_{xx}\left(\frac{\sqrt{6}}{3}, 0\right) & f''_{xy}\left(\frac{\sqrt{6}}{3}, 0\right) \\ f''_{xy}\left(\frac{\sqrt{6}}{3}, 0\right) & f''_{yy}\left(\frac{\sqrt{6}}{3}, 0\right) \end{bmatrix} = \begin{bmatrix} -2\sqrt{e} & 0 \\ 0 & \frac{2}{3}\sqrt{e} \end{bmatrix},$$

that has the eigenvalues $-2\sqrt{e}$ and $\frac{2}{3}\sqrt{e}$.

Since the two eigenvalues for $\mathbf{H}\left(\frac{\sqrt{6}}{3}, 0\right)$ are of opposite sign, then $f\left(\frac{\sqrt{6}}{3}, 0\right)$ is neither a proper local minimum nor a proper local maximum.

Problem 3

```
> restart:with(LinearAlgebra):with(plots):
> prik:=(x,y)->VectorCalculus[DotProduct](x,y):
```

```
> kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector)
  :
```

A cylindrical surface F in the (x, y, z)-space is given by the parametric representation

```
> r:=(u,v)-><u^2-u,u^2+u,v*u>:
> r(u,v);
```

$$\begin{bmatrix} u^2 - u \\ u^2 + u \\ v\,u \end{bmatrix} \quad \textbf{(3.1)}$$

where $u \in \left[0; \frac{\sqrt{3}}{2}\right]$ and $v \in [0; 1]$.

```
> plot3d(r(u,v),u=0..sqrt(3)/2,v=0..1,orientation=[-40,80],axes=
  normal,view=[-0.3..0,0..2,-0.01..1]);
```



Question 1

```
> ru:=diff~(r(u,v),u);
```

$$ru := \begin{bmatrix} 2\,u - 1 \\ 2\,u + 1 \\ v \end{bmatrix} \quad \textbf{(3.1.1)}$$

```
> rv:=diff~(r(u,v),v);
```

$$rv := \begin{bmatrix} 0 \\ 0 \\ u \end{bmatrix} \quad \textbf{(3.1.2)}$$

The normal vector is

```
> N:=kryds(ru,rv);
```

$$N := \begin{bmatrix} (2u+1)u \\ -(2u-1)u \\ 0 \end{bmatrix} \quad \textbf{(3.1.3)}$$

The Jacobi function corresponding to **r** is

```
> Jacobi:=simplify(sqrt(prik(N,N)));
```

$$Jacobi := \sqrt{2}\sqrt{4u^4+u^2} \quad \textbf{(3.1.4)}$$

$$\mathrm{Ar}(F) = \int_F 1 \, \mathrm{d}\mu = \int_0^1 \int_0^{\frac{\sqrt{3}}{2}} \mathrm{Jacobi}(u, v) \, \mathrm{d}u \, \mathrm{d}v$$

```
> Int(Int(Jacobi,u=0..sqrt(3)/2),v=0..1)=int(int(Jacobi,u=0..
  sqrt(3)/2),v=0..1);
```

$$\int_0^1 \int_0^{\frac{1}{2}\sqrt{3}} \sqrt{2}\sqrt{4u^4+u^2} \, \mathrm{d}u \, \mathrm{d}v = \frac{7}{12}\sqrt{2} \quad \textbf{(3.1.5)}$$

Let L denote the directrix in the (x,y)-plane corresponding to F.

Question 2

L is the curve of intersection between the surface F and the (x, y)-plane.
a parametric representation for L is then

```
> s:=r(u,0);
```

$$s := \begin{bmatrix} u^2-u \\ u^2+u \\ 0 \end{bmatrix} \quad \textbf{(3.2.1)}$$

where $u \in \left[0, \frac{\sqrt{3}}{2}\right]$.

```
> su:=diff~(s,u);
```

$$su := \begin{bmatrix} 2u-1 \\ 2u+1 \\ 0 \end{bmatrix} \quad \textbf{(3.2.2)}$$

The Jacobi function corresponding to L hørende Jacobifunktion is

```
> Jacobi:=simplify(sqrt(prik(su,su)));
```

$$Jacobi := \sqrt{8u^2+2} \quad \textbf{(3.2.3)}$$

Question 3

$$\int_L \frac{1}{2}(y - x)\, \mathrm{d}\mu = \int_0^{\frac{\sqrt{3}}{2}} \frac{1}{2}\,(y(u) - x(u))\mathrm{Jacobi}(u)\, \mathrm{d}u = \int_0^{\frac{\sqrt{3}}{2}} \mathrm{u\ Jacobi}(u)\, \mathrm{d}u$$

```
> Int(u*Jacobi,u=0..sqrt(3)/2)=int(u*Jacobi,u=0..sqrt(3)/2);
```

$$\int_0^{\frac{1}{2}\sqrt{3}} u\sqrt{8\,u^2 + 2}\, \mathrm{d}u = \frac{7}{12}\sqrt{2} \qquad \textbf{(3.3.1)}$$

Problem 4

```
> restart:with(LinearAlgebra):with(plots):
> prik:=(x,y)->VectorCalculus[DotProduct](x,y):
> kryds:=(x,y)->convert(VectorCalculus[CrossProduct](x,y),Vector)
  :
> div:=V->VectorCalculus[Divergence](V):
> rot:=proc(X) uses VectorCalculus;BasisFormat(false);Curl(X)end
  proc:
```

A vector field in the (x, y, z)-space is given by

```
> V:=(x,y,z)-><x^2,-2*y*x,z>:
> V(x,y,z);
```

$$\begin{bmatrix} x^2 \\ -2\,y\,x \\ z \end{bmatrix} \qquad \textbf{(4.1)}$$

In the (x, z)-plane we consider a profile region A given by the parametric representation

```
> s:=(u,v)-><u,0,v*sin(u)>:
> s(u,v);
```

$$\begin{bmatrix} u \\ 0 \\ v\sin(u) \end{bmatrix} \qquad \textbf{(4.2)}$$

where $u \in \left[0;\ \frac{1}{2}\pi\right]$ and $v \in [0;\ 1]$.

```
> plot3d(s(u,v),u=0..Pi/2,v=0..1,axes=normal,scaling=constrained)
  ;
```



A solid of revolution Ω appears by rotation of A through the angle $\frac{\pi}{4}$ about the z-axis counter-clockwise as seen from the positive end of the z-axis.

Question 1

The parametric representation for Ω

```
> r:=(u,v,w)-><u*cos(w),u*sin(w),v*sin(u)>:
> r(u,v,w);
```

$$\begin{bmatrix} u\cos(w) \\ u\sin(w) \\ v\sin(u) \end{bmatrix} \qquad \textbf{(4.1.1)}$$

where $u \in \left[0; \frac{1}{2}\pi\right]$, $v \in [0; 1]$ and $w \in \left[0; \frac{1}{4}\pi\right]$.

Question 2

∂Ω is the closed surface of Ω oriented with outward-pointing unit normal vector.
From Gauss' theorem we then get fås da

$$\text{Flux}(\mathbf{V}, \partial\Omega) = \int_{\Omega} \text{Div}(\mathbf{V})\, d\mu = \int_0^{\frac{\pi}{4}} \int_0^1 \int_0^{\frac{\pi}{2}} \text{Div}(\mathbf{V})(\mathbf{r}(u, v, w))\, \text{Jacobi}(u, v, w)\, du\, dv\, dw$$

```
> div(V)(x,y,z);
```

$$1 \tag{4.2.1}$$

```
> M:=<diff~(r(u,v,w),u)|diff~(r(u,v,w),v)|diff~(r(u,v,w),w)>;
```

$$M := \begin{bmatrix} \cos(w) & 0 & -u\sin(w) \\ \sin(w) & 0 & u\cos(w) \\ v\cos(u) & \sin(u) & 0 \end{bmatrix} \tag{4.2.2}$$

```
> Jr:=simplify(Determinant(M));
```

$$Jr := -\sin(u)\, u \tag{4.2.3}$$

that is ≤ 0, since $u \in \left[0; \frac{1}{2}\right]$. The jacobi function corresponding to **r** is then

```
> Jacobi:=-Jr;
```

$$Jacobi := \sin(u)\, u \tag{4.2.4}$$

```
> integranden:=1*Jacobi;
```

$$integranden := \sin(u)\, u \tag{4.2.5}$$

```
> Int(Int(Int(integranden,u=0..Pi/2),v=0..1),w=0..Pi/4)=int(int
  (int(integranden,u=0..Pi/2),v=0..1),w=0..Pi/4);
```

$$\int_0^{\frac{1}{4}\pi} \int_0^1 \int_0^{\frac{1}{2}\pi} \sin(u)\, u\, du\, dv\, dw = \frac{1}{4}\pi \tag{4.2.6}$$

Question 3

Let G denote the part of the surface of Ω, that bounds Ω above. Thus G is the surface of revolution that appears by rotating the upper boundary curve for A in the (x, z)-plane throung the angle $\frac{\pi}{4}$ about the z-axis counter-clockwize as seen from the poitive side of the z-axis.

A parametric representation G is then

```
> R:=r(u,1,w);
```

$$R := \begin{bmatrix} u\cos(w) \\ u\sin(w) \\ \sin(u) \end{bmatrix} \tag{4.3.1}$$

where $u \in \left[0; \frac{1}{2}\pi\right]$ and $w \in \left[0; \frac{1}{4}\pi\right]$.

```
> Ru:=diff~(R,u);
```

$$Ru := \begin{bmatrix} \cos(w) \\ \sin(w) \\ \cos(u) \end{bmatrix} \tag{4.3.2}$$

```
> Rw:=diff~(R,w);
```

(4.3.3)

$$Rw := \begin{bmatrix} -u\sin(w) \\ u\cos(w) \\ 0 \end{bmatrix} \quad \textbf{(4.3.3)}$$

Thenormal vector of the surface is

```
> N:=simplify(kryds(Ru,Rw));
```

$$N := \begin{bmatrix} -\cos(u)\,u\cos(w) \\ -\cos(u)\,u\sin(w) \\ u \end{bmatrix} \quad \textbf{(4.3.4)}$$

$\mathbf{N}(u, w)$ forms an acute angle with the z-axis, since its z-coordinat $z(u, w) = u > 0$ for $u \in \left]0; \frac{1}{2}\pi\right]$.

```
> flade:=plot3d(R,u=0..Pi/2,w=0..Pi/4,scaling=constrained):
> pil:=arrow(subs(u=1,w=Pi/8,R),subs(u=1,w=Pi/8,N)):
> display(flade,pil,orientation=[-100,70]);
```



With the chosen orientation of the closed boundary curve ∂G for G we then have

$\mathbf{n}_G = \dfrac{\mathbf{N}(u, w)}{|\mathbf{N}(u, w)|}$ for $u \in \left]0; \frac{1}{2}\pi\right]$ (the right-hand rule).

From Stokes' theorem we then have

$$\text{Cirk}(\mathbf{V}, \partial G) = \text{Flux}(\mathbf{Curl}(\mathbf{V}), G) = \int_G \mathbf{n}_G \cdot \mathbf{Curl}(\mathbf{V})\, d\mu = \int_0^{\frac{\pi}{4}} \int_0^{\frac{\pi}{2}} \mathbf{N}(u, w) \cdot \mathbf{Curl}(\mathbf{V})(\mathbf{R}(u,w))\, du\, dw$$

```
> rot(V)(x,y,z);
```

$$\begin{bmatrix} 0 \\ 0 \\ -2\,y \end{bmatrix} \quad \textbf{(4.3.5)}$$

The curl of the vector field on the surface is

```
> Curl:=<0,0,-2*u*sin(w)>;
```

$$Rot := \begin{bmatrix} 0 \\ 0 \\ -2\,u\sin(w) \end{bmatrix} \quad \textbf{(4.3.6)}$$

```
> integranden:=prik(N,Curl);
```

$$integranden := -2\,u^2 \sin(w) \quad \textbf{(4.3.7)}$$

```
> Int(Int(integranden,u=0..Pi/2),w=0..Pi/4)=int(int
  (integranden,u=0..Pi/2),w=0..Pi/4);
```

$$\int_0^{\frac{1}{4}\pi} \int_0^{\frac{1}{2}\pi} \left(-2\,u^2 \sin(w)\right) du\, dw = -\frac{1}{12}\pi^3 + \frac{1}{24}\sqrt{2}\,\pi^3 \quad \textbf{(4.3.8)}$$